



cIGdiplus

Tutoriel : Personnaliser la forme des formulaires Module VBA de gestion d'image

par Thierry GASPHERMENT (arkham46.developpez.com/)

Date de publication : 24/04/08

Dernière mise à jour : 24/04/08

Ce tutoriel explique les méthodes pour créer des formulaires **Access** aux formes personnalisées.

La librairie Gdi+ ainsi que le module VBA cIGdiPlus sont requis.


I - Introduction.....	3
II - Prérequis Gdi+.....	3
III - Initialisation des formulaires.....	4
IV - Formulaire personnalisé dont la forme est délimitée par des contrôles.....	4
IV-A - Création des contrôles.....	4
IV-B - Bouton de fermeture du formulaire.....	4
IV-C - Limitation de l'affichage au contrôle FrameVisible.....	5
IV-C - Déplacer le formulaire.....	5
IV-D - Faire un trou dans le formulaire.....	6
V - Formulaire personnalisé dont la forme est délimitée par une image.....	7
IV-A - Préparation de l'image.....	7
IV-B - Création des contrôles.....	7
IV-C - Bouton de fermeture du formulaire.....	7
IV-D - Limitation de l'affichage à l'intérieur de l'image.....	8
IV-E - Déplacer le formulaire.....	9
V - Conclusion.....	9
VI - Téléchargements.....	9

I - Introduction

Ce tutoriel a été rédigé avec Access 2003.

Nous allons au cours de ce tutoriel :

- limiter l'affichage du formulaire à l'intérieur d'un contrôle
- faire un trou dans un formulaire
- créer un formulaire en forme de bulle



 **Consultez la documentation des fonctions et des propriétés**

 **Consultez le tutoriel d'initiation**

 **Lien vers la librairie en téléchargement sur Microsoft.com pour Windows autre que XP**

II - Prérequis Gdi+

Téléchargez la classe au format texte (HTTP)

Créez un **nouveau module de classe**.

Collez-y le contenu du fichier texte.

Sauvegardez le module avec le nom **CIGdiPlus**.

Attention : le nom sous lequel vous sauvegardez le module est important.

Ce module de classe VBA utilise la librairie gdiplus.dll de Microsoft.

Si vous utilisez une version de Windows autre que XP, **télécharger la librairie** et placez la dans le même répertoire que le fichier Access.

III - Initialisation des formulaires

Cette première étape est commune et sera à répéter pour chaque exemple. Créez un nouveau formulaire et insérez-y le code suivant.

```

Initialisation du formulaire

Option Compare Database
Option Explicit
Private clGdip As clGdiPlus ' Classe image

'-----
' Sur chargement du formulaire
'-----

Private Sub Form_Load()
' Initialisation de la classe de dessin
Set clGdip = New clGdiPlus
End Sub

'-----
' Sur fermeture du formulaire
'-----

Private Sub Form_Close()
' Libération de la classe à la fermeture du formulaire
If Not clGdip is Nothing Then Set clGdip = Nothing
End Sub
    
```

Cette étape est expliquée en détail dans **le tutoriel d'initiation**.


IV - Formulaire personnalisé dont la forme est délimitée par des contrôles

Initialiser un nouveau formulaire comme décrit en section **III - Initialisation des formulaires**.

IV-A - Création des contrôles

Pour préparer le formulaire, créez :

- un rectangle nommé **FrameVisible** : ne sera visible que l'intérieur de ce rectangle
- un rectangle nommé **FrameEmpty** : il y aura un "trou" à la place de ce rectangle
- divers contrôles dont un bouton de commande nommé **BtnClose**



IV-B - Bouton de fermeture du formulaire

Comme nous allons restreindre la zone d'affichage, la barre de titre du formulaire ne sera plus visible. Il faut donc penser à ajouter un bouton de fermeture; c'est le bouton **BtnClose**.

Code de fermeture

```

' -----
' Bouton de fermeture du formulaire
' -----
Private Sub BtnClose_Click()
DoCmd.Close acForm, Me.Name
End Sub

```

IV-C - Limitation de l'affichage au contrôle FrameVisible

Pour restreindre l'affichage d'un formulaire, il faut créer une région puis l'affecter au formulaire. Nous effectuons cette action au chargement du formulaire, dans l'événement "Sur chargement", à la suite de l'initialisation de la classe.

La région rectangulaire est créée à l'aide de la fonction **CreateRegionRect**.

Les fonctions **PointsToPixelsX** et **PointsToPixelsY** sont nécessaire pour convertir les coordonnées de Twips en Pixels.

Création de la région

```

' Création d'une région rectangulaire correspondant au contrôle FrameVisible
clGdip.CreateRegionRect "regionvisible", _
    clGdip.PointsToPixelsX(Me.FrameVisible.Left) - 1, _
    clGdip.PointsToPixelsY(Me.FrameVisible.Top) - 1, _
    clGdip.PointsToPixelsX(Me.FrameVisible.Left + Me.FrameVisible.Width) - 1, _
    clGdip.PointsToPixelsY(Me.FrameVisible.Top + Me.FrameVisible.Height) - 1

```

La région est affectée au formulaire à l'aide de la fonction **SetFormRegion**.

On précise le formulaire (**Me** pour le formulaire dans lequel le code s'exécute) et le nom de la région précédemment créée.

On peut ensuite supprimer la région qui n'est plus utile.

Remarque : on ne précise pas de contrôle de référence (**pCtrlRef**) car la région n'a pas été créée sur une image mais directement avec des coordonnées prises sur le formulaire.


Affectation de la région au formulaire

```

' Affectation de la région au formulaire
clGdip.SetFormRegion Me,"regionvisible"
' Supprime la région
clGdip.DeleteRegion "regionvisible"

```

Lorsqu'on affiche alors le formulaire, on ne voit que l'intérieur du contrôle FrameVisible.



IV-C - Déplacer le formulaire

Si vous essayer de déplacer le formulaire, vous êtes bien embêtés car la barre de titre a disparu.

Utilisez la fonction **DragForm** sur l'événement **Souris Appuyée** pour permettre le déplacement en cliquant sur le formulaire.

Déplacer le formulaire en cliquant sur la section détail

```

'-----
' Souris appuyée sur section détail
'-----
Private Sub Détail_MouseDown(Button As Integer, Shift As Integer, X As Single, Y As Single)
    clGdip.DragForm Me
End Sub
    
```

IV-D - Faire un trou dans le formulaire

Pour faire un trou à l'emplacement du contrôle **FrameEmpty** nous allons :


- créer une région correspondant à ce contrôle
- combiner la région avec la précédente
- Affecter la région résultant de la combinaison au formulaire

Faire un trou dans le formulaire

```

'-----
' Sur chargement du formulaire
'-----
Private Sub Form_Load()
' Initialisation de la classe de dessin
Set clGdip = New ClGdiPlus
' Création d'une région rectangulaire correspondant au contrôle FrameVisible
clGdip.CreateRegionRect "regionvisible", _
    clGdip.PointsToPixelsX(Me.FrameVisible.Left) - 1, _
    clGdip.PointsToPixelsY(Me.FrameVisible.Top) - 1, _
    clGdip.PointsToPixelsX(Me.FrameVisible.Left + Me.FrameVisible.Width) - 1, _
    clGdip.PointsToPixelsY(Me.FrameVisible.Top + Me.FrameVisible.Height) - 1
' Création d'une région rectangulaire correspondant au contrôle FrameEmpty
clGdip.CreateRegionRect "regionempty", _
    clGdip.PointsToPixelsX(Me.FrameEmpty.Left), _
    clGdip.PointsToPixelsY(Me.FrameEmpty.Top), _
    clGdip.PointsToPixelsX(Me.FrameEmpty.Left + Me.FrameEmpty.Width) - 2, _
    clGdip.PointsToPixelsY(Me.FrameEmpty.Top + Me.FrameEmpty.Height) - 2
' Exclu la region "regionempty" de la région "regionvisible"
clGdip.RegionCombine "regionvisible", "regionempty", CombineModeExclude
' Affectation de la région au formulaire
clGdip.SetFormRegion Me, "regionvisible"
' Suppression des régions
clGdip.DeleteRegion "regionvisible"
clGdip.DeleteRegion "regionempty"
End Sub
    
```

Lorsqu'on affiche alors le formulaire, on ne voit que l'intérieur du contrôle **FrameVisible** avec un trou à l'emplacement du contrôle **FrameEmpty**.



V - Formulaire personnalisé dont la forme est délimitée par une image

Initialiser un nouveau formulaire comme décrit en section III - **Initialisation des formulaires**.

IV-A - Préparation de l'image

Créez une image de fond, avec **Paint** par exemple.

Laissez le tour de l'image en blanc, cette couleur délimitera l'affichage.



IV-B - Création des contrôles

Pour préparer le formulaire, créez :

- une image nommée **ImgBack** qui contient l'image précédemment créée.
- une étiquette contenant un texte libre
- un bouton de commande nommé **BtnClose**



IV-C - Bouton de fermeture du formulaire

Comme nous allons restreindre la zone d'affichage, la barre de titre du formulaire ne sera plus visible.

Il faut donc penser à ajouter un bouton de fermeture; c'est le bouton **BtnClose**.

Code de fermeture

```

' -----
' Bouton de fermeture du formulaire
' -----
Private Sub BtnClose_Click()
DoCmd.Close acForm, Me.Name
End Sub

```

IV-D - Limitation de l'affichage à l'intérieur de l'image

Pour restreindre l'affichage d'un formulaire, il faut créer une région puis l'affecter au formulaire. Nous effectuons cette action au chargement du formulaire, dans l'événement "Sur chargement", à la suite de l'initialisation de la classe.

La première étape est le chargement de l'image intégrée au contrôle.

Nous utilisons la fonction **LoadBitmapFromControl** pour charger l'image du contrôle **ImgBack**.

La région est créée à l'aide de la fonction **CreateRegionFromColor**.

Cette fonction crée une région définie par la couleur des points de l'image.

On utilise la fonction **GetPixel** pour lire la couleur du point en haut à gauche (coordonnées 0,0).

Création de la région

```

' Chargement de l'image en mémoire à partir du contrôle
clGdip.LoadBitmapFromControl Me.ImgBack
' Créaton d'une région définie par les points dont la couleur est différente du point de coordonnées 0,0
clGdip.CreateRegionFromColor "region", clGdip.GetPixel(0, 0)

```

La région est affectée au formulaire à l'aide de la fonction **SetFormRegion**.

On précise le formulaire (**Me** pour le formulaire dans lequel le code s'exécute) et le nom de la région précédemment créée.

Les valeurs numériques sont des corrections en pixels.

Les corrections nécessaires à un affichage correct peuvent différer en fonction de la taille et de la position du contrôle et de l'image intégrée.

On peut ensuite supprimer la région qui n'est plus utile.

Remarque : on précise un contrôle de référence (**pCtrlRef**) car la région a été créée par rapport à l'image contenue dans le contrôle **ImgBack**.


Affectation de la région au formulaire

```

' Affectation de la région au formulaire
clGdip.SetFormRegion Me, "region", Me.ImgBack, , -2, -3, , 1.5
' Supprime la région
clGdip.DeleteRegion "region"

```

Lorsqu'on affiche alors le formulaire, on ne voit que l'intérieur de la bulle.



IV-E - Déplacer le formulaire

Si vous essayer de déplacer le formulaire, vous êtes bien embêtés car la barre de titre a disparu. Utilisez la fonction **DragForm** sur l'événement **Souris Appuyée** pour permettre le déplacement en cliquant sur le formulaire.

Déplacer le formulaire en cliquant sur le contrôle image

```
'-----  
' Souris appuyée sur l'image  
'-----  
Private Sub ImgBack_MouseDown(Button As Integer, Shift As Integer, X As Single, Y As Single)  
    clGdip.DragForm Me  
End Sub
```

V - Conclusion

Merci à l'équipe Office de developpez.com pour ses relectures, commentaires et encouragements!

VI - Téléchargements

Télécharger la base Access de ce tutoriel au format ACCESS 2000 ([HTTP](#))